# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: Blair Bonnett <blair dot bonnett at gmail dot com>

_name=sphinx-copybutton
pkgname=python-$_name
pkgver=0.5.2
pkgrel=1
pkgdesc="Sphinx extension to add a \"copy\" button to code blocks"
arch=('any')
url="https://github.com/executablebooks/sphinx-copybutton"
license=('MIT')
depends=('python-sphinx')
makedepends=('python-setuptools')
source=(https://files.pythonhosted.org/packages/source/s/$_name/$_name-$pkgver.tar.gz)
sha256sums=('4cf17c82fb9646d1bc9ca92ac280813a3b605d8c421225fd9913154103ee1fbd')

build() {
  cd $_name-$pkgver
  python setup.py build
}

package() {
  cd $_name-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
  install -Dm644 -t "$pkgdir/usr/share/licenses/$pkgname" LICENSE
  install -Dm644 -t "$pkgdir/usr/share/doc/$pkgname" README.md
}

# vim:set ts=2 sw=2 et:
